```
model{
#Model for log-hazard ratios
for(i in 1:ndp){
        prec[i]<- 1/(se[i]*se[i])
        lhr[i]~dnorm(delta[i],prec[i])

#Random effects model for log hazard ratios
                delta[i] ~ dnorm(md[i],taud[i])
                taud[i] <- tau * (1 + equals(arm[i],3) /3)
                md[i] <- d[t[i]] - d[b[i]] + equals(arm[i],3) * sw[i]
#Calculation of residual deviance
        rhat[i] <- lhr[i] * prec[i]
        dev[i] <- (lhr[i] - delta[i])*(lhr[i] - delta[i])/(se[i]*se[i])
        }
        resdev <- sum(dev[])

# Adjustment for multi-arm trials
        sw[1]<- 0
        for (i in 2:ndp) { sw[i] <- (delta[i-1] - d[t[i-1]] + d[b[i-1]])/2}

#Non-informative priors for log hazard ratios
        d[1]<- 0
        for (k in 2:nt){
        d[k] ~ dnorm(0,.00001) # vague priors for basic parameters

        }

        sd~dunif(0,100)
        tau<-1/pow(sd,2)

#Rank the treatment effects (with 1=best) & record the best treatment
for(k in 1:nt){
        rk[k]<- rank(d[],k)

        best[k]<-equals(rk[k],1)
        }

#All pair-wise log hazard ratios and hazard ratios
for (c in 1:nt-1){
        for (k in (c+1):nt){
                lhzr[c,k] <- d[k] - d[c]
                HR[c,k] <- exp(lhzr[c,k])
                }
        }
}
```